

AperTO - Archivio Istituzionale Open Access dell'Università di Torino

Certifying Algorithms and Relevant Properties of Reversible Primitive Permutations with Lean

This is the author's manuscript

Original Citation:

Availability:

This version is available <http://hdl.handle.net/2318/1867901> since 2022-06-30T11:05:16Z

Publisher:

Springer International Publishing

Published version:

DOI:10.1007/978-3-031-09005-9_8

Terms of use:

Open Access

Anyone can freely access the full text of works made available as "Open Access". Works made available under a Creative Commons license can be used according to the terms and conditions of said license. Use of all other works requires consent of the right holder (author or publisher) if not exempted from copyright protection by the applicable law.

(Article begins on next page)

Certifying algorithms and relevant properties of Reversible Primitive Permutations with Lean

Giacomo Maletto¹ and Luca Roversi²[0000–0002–1871–6109]

¹ Università degli Studi di Torino, Dipartimento di Matematica – Italy
`giacomo.maletto@edu.unito.it`

² Università degli Studi di Torino, Dipartimento di Informatica – Italy
`luca.roversi@unito.it`

Abstract. Reversible Primitive Permutations (RPP) are recursively defined functions designed to model Reversible Computation. We illustrate a proof, fully developed with the proof-assistant `Lean`, certifying that: “RPP can encode every Primitive Recursive Function”. Our reworking of the original proof of that statement is conceptually simpler, fixes some bugs, suggests a new more primitive reversible iteration scheme for RPP, and, in order to keep formalization and semi-automatic proofs simple, led us to identify a single pattern that can generate some useful reversible algorithms in RPP: Cantor Pairing, Quotient/Reminder of integer division, truncated Square Root. Our `Lean` source code is available for experiments on Reversible Computation whose properties can be certified.

1 Introduction

Studies focused on questions posed by Maxwell, regarding the solidity of the principles which Thermodynamics is based on, recognized the fundamental role that Reversible Computation can play to that purpose.

Once identified, it has been apparent that Reversible Computation constitutes the context in which to frame relevant aspects in areas of Computer Science; they can span from reversible hardware design which can offer a greener foot-print, as compared to classical hardware, to unconventional computational models — we think of quantum or bio-inspired ones, for example —, passing through parallel computation and the synchronization issues that it rises, or debuggers that help tracing back to the origin of a bug, or the consistent transactions roll-back in data-base management systems, just to name some. The book [18] is a comprehensive introduction to the subject; the book [6], focused on the low-level aspects of Reversible Computation, concerning the realization of reversible hardware, and [13], focused on how models of Reversible Computation like Reversible Turing Machines (RTM), and Reversible Cellular Automata (RCA) can be considered universal and how to prove that they enjoy such a property, are complementary to, and integrate [18].

This work focuses on the *functional model* RPP [17] of Reversible Computation. RPP stands for (the class of) Reversible Primitive Permutations, which can be seen as a possible reversible counterpart of PRF, the class of Primitive

Recursive functions [19]. We recall that RPP, in analogy with PRF, is defined as the smallest class built on some given basic reversible functions, closed under suitable composition schemes. The very functional nature of the elements in RPP is at the base of reasonably accessible proofs of the following properties:

- RPP is PRF-complete [17]: for every function $F \in \text{PRF}$ with arity $n \in \mathbb{N}$, both $m \in \mathbb{N}$ and \mathbf{f} in RPP exist such that \mathbf{f} encodes F , i.e. $\mathbf{f}(z, \bar{x}, \bar{y}) = (z + F(\bar{x}), \bar{x}, \bar{y})$, for every $\bar{x} \in \mathbb{N}^n$, whenever all the m variables in \bar{y} are set to the value 0. Both z and the tuple \bar{y} are *ancillae*. They can be thought of as temporary storage for intermediate computations of the encoding.
- RPP can be extended to become Turing-complete [16] by means of a minimization scheme analogous to the one that extends PRF to the Turing-complete class of *Partial* Recursive Functions.
- According to [12], RPP and the reversible programming language SRL [11] are equivalent, so the fix-point problem is undecidable for RPP as well [10].

This work is further evidence that expressing Reversible Computation by means of recursively defined computational models like RPP, *naturally* offers the possibility to certify with *reasonable effort* the correctness, or other interesting properties, of algorithms in RPP, by means of some proof-assistant, also discovering new algorithms. We recall that a proof-assistant is an integrated environment to formalize data-types, to implement algorithms on them, to formalize specifications and prove that they hold, increasing algorithms dependability.

Contributions. We show how to express RPP and its evaluation mechanism inside the proof-assistant Lean [5]. We can certify the correctness of every reversible function of RPP with respect to a given specification which also means certifying that RPP is PRF-complete, the main result in [17]. In more detail:

- we give a strong guarantee that RPP is PRF-complete in three macro steps. We exploit that in Lean `mathlib` library, PRF is proved equivalent to a class of recursive *unary* functions called `primrec`. We define a data-type `rpp` in Lean to represent RPP. Then, we certify that, for any function `f:primrec`, i.e. any unary `f` with type `primrec` in Lean, a function exists with type `rpp` that encodes `f:primrec`. Apart from fixing some bugs, our proof is fully detailed as compared to [17]. Moreover it’s conceptually and technically simpler;
- concerning simplification, it follows from how the elements in `primrec` work, and, additionally, it is characterized by the following aspects:
 - we define a *new* finite reversible iteration scheme subsuming the reversible iteration schemes in RPP, and SRL, but which is more primitive;
 - we identify an algorithmic pattern which uniquely associates elements of \mathbb{N}^2 , and \mathbb{N} by counting steps in specific paths. The pattern becomes a reversible element in `rpp` once fixed the parameter it depends on. Slightly different parameter instances generate reversible algorithms whose behavior we can certify in Lean. They are truncated Square Root, Quotient/Reminder of integer division, and Cantor Pairing [2,20]. The original proof in [17] that RPP is PRF-complete relies on Cantor Pairing, used as

a stack to keep the representation of a PRF function as element of RPP reversible. Our proof in Lean replaces Cantor Pairing with a reversible representation of functions `mkpair/unpair` that `mathlib` supplies as isomorphism $\mathbb{N} \times \mathbb{N} \simeq \mathbb{N}$. The truncated Square Root is the basic ingredient to obtain reversible `mkpair/unpair`.

Related work. Concerning the formalization in a proof-assistant of the semantics, and its properties, of a formalism for Reversible Computation, we are aware of [15]. By means of the proof-assistant `Matita` [1], it certifies that a denotational semantics for the imperative reversible programming language `Janus` [18, Section 8.3.3] is fully abstract with respect to the operational semantics.

Concerning *functional models* of Reversible Computation, we are aware of [7] which introduces the class of reversible functions `RI`, which is as expressive as the *Partial* Recursive Functions. So, `RI` is stronger than `RPP`, however we see `RI` as less abstract than `RPP` for two reasons: (i) the primitive functions of `RI` depend on a given specific binary representation of natural numbers; (ii) unlike `RPP`, which we can see as `PRF` in a reversible setting, it is not evident to us that `RI` can be considered the natural extension of a total class analogous to `RPP`.

Contents. This work illustrates the relevant parts of the BSc Thesis [8] which comes with [9], a Lean project that certifies properties, and algorithms of `RPP`. Section 2 recalls the class `RPP` by commenting on the main design aspects that characterize its definition inside Lean. Section 3 defines and proves correct new reversible algorithms central to the proof. Section 4 recalls the main aspects of `primrec`, and illustrates the key steps to port the original `PRF`-completeness proof of `RPP` to Lean. Section 5 is about possible developments.

2 Reversible Primitive Permutations (RPP)

We use the data-type `rpp` in Figure 1, as defined in Lean, to recall from [17] that the class `RPP` is the smallest class of functions that contains five base functions, named as in the definition, and all the functions that we can generate by the composition schemes whose name is next to the corresponding clause in Figure 1. For ease of use and readability the last two lines in Figure 1 introduce infix notations for series and parallel compositions.

Example 1 (A term of type rpp). In `rpp` we can write `(Id 1||Sw);;(It Su)||(Id 1);;(Id 1||If Su (Id 1) Pr)` which we also represent as a diagram. Its inputs are the names to the left of the blocks. The outputs are to their right:

$$\begin{array}{c}
 x \text{ Id } 1 \quad x \quad \text{It Su} \quad x \quad \text{Id } 1 \quad x \\
 y \quad \text{Sw} \quad z \quad z+x \quad \text{If Su (Id 1) Pr} \quad z+x \\
 z \quad y \quad \text{Id } 1 \quad y \quad w \quad w
 \end{array}
 \quad w = \begin{cases}
 y+1 & \text{if } z+x > 0 \\
 y & \text{if } z+x = 0 \\
 y-1 & \text{if } z+x < 0
 \end{cases} .$$

We have just built a series composition of three parallel compositions. The first one composes a unary identity `Id 1`, which leaves its unique input untouched,

```

inductive rpp : Type
  -- Base functions
  | Id (n : ℕ) : rpp -- Identity
  | Ne : rpp      -- Sign-change
  | Su : rpp      -- Successor
  | Pr : rpp      -- Predecessor
  | Sw : rpp      -- Transposition or Swap
  -- Inductively defined functions
  | Co (f g : rpp) : rpp -- Series composition
  | Pa (f g : rpp) : rpp -- Parallel composition
  | It (f : rpp) : rpp -- Finite iteration
  | If (f g h : rpp) : rpp -- Selection
infix '||' : 55 := Pa -- Notation for the Parallel composition
infix ';;' : 50 := Co -- Notation for the Series composition

```

Fig. 1: The class RPP as a data-type `rpp` in Lean.

and `Sw`, which swaps its two arguments. Then, the x -times iteration of the successor `Su`, i.e. `It Su`, is in parallel with `Id 1`: that is why, one of the outputs of `It Su` is $z + x$. Finally, `If Su (Id 1) Pr` selects which among `Su`, `Id 1`, and `Pr` to apply to the argument y , depending on the value of $z + x$; in particular, `Pr` is the function that computes the predecessor of the argument. Figure 5 will give the operational semantics which defines `rpp` formally as a class of functions on \mathbb{Z} , not on \mathbb{N} . ■

Remark 1 (“Weak weakening” of algorithms in `rpp`). We typically drop `Id m` if it is the last function of a parallel composition. For example, term and diagram in *Example 1* become `(Id 1 || Sw) ;; (It Su) ;; (Id 1 || If Su (Id 1) Pr)` and:

$$\begin{array}{c}
 x \text{ Id } 1 \ x \\
 y \text{ Sw } \ z \\
 z \text{ Sw } \ y
 \end{array}
 \begin{array}{c}
 x \text{ It } \text{Su} \\
 z + x \\
 y
 \end{array}
 \begin{array}{c}
 x \text{ Id } 1 \\
 z + x \\
 y
 \end{array}
 \begin{array}{c}
 x \\
 z + x \\
 w
 \end{array}
 \quad w = \begin{cases} y + 1 & \text{if } z + x > 0 \\ y & \text{if } z + x = 0 \\ y - 1 & \text{if } z + x < 0 \end{cases} .$$

Remark 2 explains why. ■


The function in Figure 2 computes the arity of any $f : \text{rpp}$ from the structure of f , once fixed the arities of the base functions; $f.\text{arity}$ is Lean dialect for the more typical notation “ $\text{arity}(f)$ ”.

Figure 3 remarks that `rpp` considers n -ary identities `Id n` as primitive; in RPP the function `Id n` is obtained by parallel composition of n unary identities.

For any given $f : \text{rpp}$, the function `inv` in Figure 4 builds an element with type `rpp`. The definition of `inv` lets the successor `Su` be inverse of the predecessor `Pr` and lets every other base function be self-dual. Moreover, the function `inv` distributes over finite iteration `It`, selection `If`, and parallel composition `||`, while it requires to exchange the order of the arguments before distributing over

```

def arity : rpp → ℕ
| (Id n)   := n
| Ne       := 1
| Su       := 1
| Pr       := 1
| Sw       := 2
| (f || g) := f.arity + g.arity
| (f ;; g) := max f.arity g.arity
| (It f)   := 1 + f.arity -- It f has an extra argument compared to f
| (If f g h) := 1 + max (max f.arity g.arity) h.arity
    
```

 Fig. 2: Arity of every $f : \text{rpp}$.

 Fig. 3: n -ary identities are base functions of rpp .

```

def inv : rpp → rpp
| (Id n)   := Id n -- self-dual
| Ne       := Ne   -- self-dual
| Su       := Pr
| Pr       := Su
| Sw       := Sw   -- self-dual
| (f || g) := inv f || inv g
| (f ;; g) := inv g ;; inv f
| (It f)   := It (inv f)
| (If f g h) := If (inv f) (inv g) (inv h)
notation f '⁻¹' := inv f
    
```

 Fig. 4: Inverse $\text{inv } f$ of every $f : \text{rpp}$.

the series composition `;;`. The last line with `notation` suggests that f^{-1} is the inverse of f ; we shall prove this fact once given the operational semantics of `rpp`.

```

def ev : rpp → list ℤ → list ℤ
| (Id n)   X           := X
| Ne      (x :: X)    := -x :: X
| Su      (x :: X)    := (x + 1) :: X
| Pr      (x :: X)    := (x - 1) :: X
| Sw      (x :: y :: X) := y :: x :: X
| (f ;; g) X         := ev g (ev f X)
| (f || g) X         := ev f (take f.arity X) ++ ev g (drop f.arity X)
| (It f)  (x :: X)    := x :: ((ev f) ^ [↓x] X)
| (If f g h) (0 :: X) := 0 :: ev g X
| (If f g h) ((n : ℕ) + 1) :: X := (n + 1) :: ev f X
| (If f g h) (-[1 + n] :: X) := -[1 + n] :: ev h X
| _       X           := X
notation '<' f '>' := ev f

```

Fig. 5: Operational semantics of elements in `rpp`.

Operational semantics of `rpp`. The function `ev` in Figure 5 interprets an element of `rpp` as a function from a list of integers to a list of integers. Originally, in [17], RPP is a class of functions with type $\mathbb{Z}^n \rightarrow \mathbb{Z}^n$. We use `list ℤ` in place of tuples of \mathbb{Z} to exploit Lean library `mathlib` and save a large amount of formalization.

Let us give a look at the clauses in Figure 5. `Id n` leaves the input list `X` untouched. `Ne` “negates”, i.e. takes the opposite sign of, the head of the list, `Su` increments, and `Pr` decrements it. `Sw` is the transposition, or swap, that exchanges the first two elements of its argument. The series composition `f;;g` first applies `f` and then `g`. The parallel composition `f||g` splits `X` into two parts. The “topmost” one (`take f.arity X`) has as many elements as the arity of `f`; the “lowermost” one (`drop f.arity X`) contains the part of `X` that can supply the arguments to `g`. Finally, it concatenates the two resulting lists by the append `++`. Our *new finite iteration* `It f` iterates `f` as many times as the value of the head `x` of the argument, if `x` contains a non negative value; otherwise it is the identity on the whole `x::X`. This behavior is the meaning of `(ev f) ^ [↓x]`. The selection `If f g h` chooses one among `f`, `g`, and `h`, depending on the argument head `x`: it is `g` with `x = 0`, it is `f` with `x > 0`, and `h` with `x < 0`. The last line of Figure 5 sets a handy notation for `ev`.

Remark 2 (We keep the definition of `ev` simple). Based on our definition, we can apply any `f:rpp` to any `X:list ℤ`. This is based on two observations: first, in Lean it holds:

```

theorem ev_split (f: rpp) (X: list ℤ):
<f> X = (<f> (take f.arity X)) ++ drop f.arity X

```

so that if $X.length \geq f.arity$, i.e. X supplies enough arguments, then f operates on the first elements of X according to its arity. This justifies *Remark 1*. Second, if instead $X.length < f.arity$ holds, i.e. X has not enough elements, $f X$ has an unspecified behavior; this might sound odd, but it simplifies the certified proofs of must-have properties of `rpp`. ■

2.1 The functions `inv h` and `h` are each other inverse

Once defined `inv` in Figure 4 and `ev` in Figure 5 we can prove:

```
theorem inv_co_l (h : rpp) (X : list ℤ) : <h ;; h-1> X = X
theorem inv_co_r (h : rpp) (X : list ℤ) : <h-1 ;; h> X = X
```

certifying that h and h^{-1} are each other inverse. We start by focusing on the main details to prove `theorem inv_co_l` in Lean. The proof proceeds by (structural) induction on h , which generates 9 cases, one for each clause that defines `rpp`. One can go through the majority of them smoothly. Some comments about two of the more challenging cases follow.

Parallel composition. Let h be some parallel composition, whose main constructor is `Pa`. The step-wise proof of `inv_co_l` is:

```
<f||g;;(f||g)-1> X
  = <f||g;;f-1||g-1> X      -- by definition
(!) = <(f;;f-1)||g;;g-1> X  -- lemma pa_co_pa, arity_inv below
  = <f;;f-1>(take f.arity X) ++ <g;;g-1>(drop f.arity X)
                                     -- by definition
  = take f.arity X ++ drop f.arity X -- by ind. hyp.
  = X                                -- property of ++ (append),
```

where the equivalence (!) holds because we can prove both:

```
lemma pa_co_pa (f f' g g' : rpp) (X : list ℤ) :
  f.arity = f'.arity → <f||g ;; f'||g'> X = <(f;;f') || (g;;g')> X ,
lemma arity_inv (f : rpp) : f-1.arity = f.arity .
```

Proving `lemma arity_inv`, i.e. that the arity of a function does not change if we invert it, assures that we can prove `lemma pa_co_pa`, i.e. that series and parallel compositions smoothly distribute reciprocally. □

Iteration. Let h be a finite iterator whose main constructor is `It`. The goal to prove is $\langle It f ;; It f^{-1} \rangle x : X = x : X$ which reduces to $\langle f^{-1} \rangle^{\lfloor x \rfloor} (\langle f \rangle^{\lfloor x \rfloor} X) = X$, where, we recall, the notation $\langle f \rangle^{\lfloor x \rfloor}$ means “ $\langle f \rangle$ applied x times, if x is positive”. Luckily this last statement is both formalized as `function .left_inverse g[n] f[n]`, available in the library `mathlib` of Lean. □

To conclude, let us see how the proof of `inv_co_r` works. It does not copy-cat the one of `inv_co_l`. It relies on proving:

lemma `inv_involute` (`f : rpp`) : $(f^{-1})^{-1} = f$,

which says that applying `inv` twice is the identity, and on using `inv_co_l`:

`<f-1 ;; f > X = X` -- which, by `inv_involute`, is equivalent to
`<f-1 ;; (f-1)-1 > X = X` -- which holds because it is an
instance of `(inv_co_l f-1)` .

Remark 3 (On our simplifying choices on `ev`). A less general, but semantically more appropriate version of `inv_co_l` and `inv_co_r` could be:

theorem `inv_co_l` (`f : rpp`) (`X : list ℤ`) :
`f.arity ≤ X.length → <f ;; f-1 > X = X`
theorem `inv_co_r` (`f : rpp`) (`X : list ℤ`) :
`f.arity ≤ X.length → <f-1 ;; f > X = X`

because, recalling *Remark 2*, `f X` makes sense when `f.arity ≤ X.length`. Fortunately, the way we defined `rpp` allows us to state `inv_co_l` or `inv_co_r` in full generality with no reference to `f.arity ≤ X.length`. ■

2.2 How `rpp` differs from original RPP

The definition of `rpp` in Lean is really very close to the original RPP, but not identical. The goal is to simplify the overall task of formalization and certification. The brief list of changes follows.

- As already outlined, `It` and `If` use the head of the input list to iterate or choose: taking the head of a list with pattern matching is obvious. In [17], the last element in the input tuple drives iteration and selection of RPP.
- `Id n`, for any `n:ℕ`, is primitive in `rpp` and derived in RPP.
- Using `list ℤ → list ℤ` as the domain of the function that interprets any given element `f:rpp` avoids letting the type of `f:rpp` depend on the arity of `f`. To know the arity of `f` it is enough to invoke `arity f`. Finally, we observe that getting rid of a dependent type like, say, `rpp n`, allows us to escape situations in which we would need to compare equal but not definitionally equal types like `rpp (n+1)` and `rpp (1+n)`.
- The new finite iterator `It f (x::t) : list ℤ` *subsumes* the finite iterators `ItR` in RPP, and `for` in SRL, i.e. `It` is more primitive, equally expressive and simpler for Lean to prove that its definition is terminating.

We recall that `ItR f (x0, x1, ..., xn-2, x)` evaluates to `f(f(...f(x0, x1, ..., xn-2)...))` with `|x|` occurrences of `f`. Instead, `for(f) x` evaluates to `f(f(...f(x0, x1, ..., xn-2)...))`, with `x` occurrences of `f`, if `x > 0`; it evaluates to `f-1(f-1(...f-1(x0, x1, ..., xn-2)...))`, with `-x` occurrences of `f-1`, if `x < 0`; it behaves like the identity if `x = 0`.

We can define both `ItR` and `for` in terms of `It`:

$$\begin{aligned} \text{ItR } f &= (\text{It } f) ;; \text{Ne} ;; (\text{It } f) ;; \text{Ne} & (1) \\ \text{for}(f) &= (\text{It } f) ;; \text{Ne} ;; (\text{It } f^{-1}) ;; \text{Ne} . & (2) \end{aligned}$$

Example 2 (How does (1) work?). Whenever $x > 0$, the leftmost `It f` in (1) iterates `f`, while the rightmost one does nothing because `Ne` in the middle negates x . On the contrary, if $x < 0$, the leftmost `It f` does nothing and the iteration is performed by the rightmost iteration, because `Ne` in the middle negates x . In both cases, the last `Ne` restores x to its initial sign. But this is the behavior of `ItR`, as we wanted. ■

3 RPP algorithms central to our proofs




Fig. 6: Some useful functions of `rpp`

Figure 6 recalls definition, and behavior of some `rpp` functions in [17]. It is worth commenting on how rewiring $[i_0 \dots i_n]$ works. Let $\{i_0, \dots, i_n\} \subseteq \{0, \dots, m\}$. Let $\{j_1, \dots, j_{m-n}\}$ be the set of remaining indices $\{0, \dots, m\} \setminus \{i_0, \dots, i_n\}$ ordered such that $j_k < j_{k+1}$. By definition, $[i_0, \dots, i_n](x_0, \dots, x_m) = (x_{i_0}, \dots, x_{i_n}, x_{j_1}, \dots, x_{j_{m-n}})$, i.e. rewiring brings every input with index in $\{i_0, \dots, i_n\}$ before all the remaining inputs, preserving the order.

Figure 7 identifies the new algorithm scheme `step[_]`. Depending on how we fill the hole `[_]`, we get step functions that, once iterated, draw paths in \mathbb{N}^2 .

On top of the functions in Figures 6, and 7 we build Cantor Pairing/Unpairing, Quotient/Reminder of integer division, and truncated Square Root. It is enough to make the correct instance of `step[_]` in order to visit \mathbb{N}^2 as in Figures 8a, 8b, and 8c, respectively. The alternative pairing `mkpair` has a more complex definition, and is a necessary ingredient for the main proof.

³ Note that using our definition, the variable `n` must be non-negative in order to have the shown behavior, otherwise the function acts as the identity. This is why it's called *increment* and not *addition*.




Fig. 7: Algorithm scheme $\text{step}[_]$. The algorithm we can obtain from it depends on how we fill the hole $[_]$.




Fig. 8: Paths in \mathbb{N}^2 that generate algorithms in rpp .

Cantor (Un-)Pairing. The standard definition of Cantor Pairing $\text{cp} : \mathbb{N}^2 \rightarrow \mathbb{N}$ and Un-pairing $\text{cu} : \mathbb{N} \rightarrow \mathbb{N}^2$, two bijections one inverse of the other, is:

$$\text{cp}(x, y) = \sum_{i=1}^{x+y} i + x = \frac{(x+y)(x+y+1)}{2} + x \quad (3)$$

$$\text{cu}(n) = \left(n - \frac{i(1+i)}{2}, \frac{i(3+i)}{2} - n \right), \quad (4)$$

where $i = \left\lfloor \frac{\sqrt{8n+1}-1}{2} \right\rfloor$.

Figure 9 has all we need to define Cantor Pairing $\text{cp}:\text{rpp}$, and Un-pairing $\text{cu}:\text{rpp}$. In Figure 9a, cp_in is the natural algorithm in rpp to implement (3). As expected, the input pair (x, y) is part of cp_in output. The suffix “**in**” in the name “recalls” exactly this aspect. In order to drop (x, y) from the output of cp_in , and obtain cp as in Figure 9e, applying *Bennet’s trick*, we need cu_in^{-1} , i.e. the inverse of cu_in which is new, as compared to [17]. The intuition behind cu_in is as follows. Let us fix any point $(x, y) \in \mathbb{N}^2$. We can realize that, starting from the origin, if we follow as many steps as the value $\text{cp}(x, y)$ in Figure 8a, we stop exactly at (x, y) . The function, expressed in standard functional notation, that, given the current point (x, y) , identifies the next one to move to in the path of Figure 8a is:

$$\text{step}(x, y) = \begin{cases} (x+1, y-1) & y > 0 \\ (0, x+1) & y = 0 \end{cases}.$$

We implement $\text{step}(x, y)$ in rpp as $\text{step}[\text{Su}; ; \text{Sw}]$. Figures 9b, and 9c represent two runs of $\text{step}[\text{Su}; ; \text{Sw}]$ to give visual evidence that $\text{step}[\text{Su}; ; \text{Sw}]$ implements $\text{step}(x, y)$. Colored occurrences of y show the relevant part of the computational flow. Note that we cannot implement $\text{step}(x, y)$ by using the conditional It directly on y , because in the computation we also want to modify the value of




Fig. 9: Cantor Pairing and Un-pairing.

y . Finally, as soon as we get cu_in by iterating $step[Su;;Sw]$ as in Figure 9d, we can define cp (Figure 9e), and cu (Figure 9f). \square




Fig. 10: Quotient/Reminder and Square root.

Quotient and reminder. Let us focus on the path in Figure 8b. It starts at $(0, n)$ (with $n = 4$), and, at every step, the next point is in *direction* $(+1, -1)$. When it reaches $(n, 0)$ (with $n = 4$), instead of jumping to $(0, n + 1)$, as in Figure 8a, it lands again on $(0, n)$. The idea is to keep looping on the same diagonal. This behavior can be achieved by iterating $step[Sw||Su]$. Figure 10a shows that we are doing modular arithmetic. Globally, it takes $n + 1$ steps from $(0, n)$ to itself

by means of `step[Sw||Su]`. Specifically, if we assume we have performed m steps along the diagonal, and we are at point (x, y) , we have that $x \equiv m \pmod{n+1}$ and $0 \leq x \leq n$. So, if we increase a counter by one each time we get back to $(0, n)$ we can calculate quotient and remainder. \square

Truncated Square root. Let us focus on the path in Figure 8c. It starts at $(0, 0)$. Whenever it reaches $(x, 0)$ it jumps to $(0, x+2)$, otherwise the next point is in *direction* $(+1, -1)$. The behavior can be achieved by iterating `step[Su;;Su;;Sw||Su]` as in Figure 10b. In order to compute $\lfloor \sqrt{n} \rfloor$, besides implementing the above path, the function `step[Su;;Su;;Sw||Su]` counts in k the number of jumps occurred so far along the path. In particular, starting from $(0, 0)$, the first jump occurs in the first step; the next one in the $(1+3)$ th, then the $(1+3+5)$ th, then the $(1+3+5+7)$ th etc. Since we know that $1+3+\dots+(2k-1) = k^2$ for any k , letting n be the number of iterations (and hence the numbers of steps) we have that k is such that $k^2 \leq n < (k+1)^2$; i.e. $k = \lfloor \sqrt{n} \rfloor$. \square

Remark 4. The value $2 \lfloor \sqrt{n} \rfloor - r$ can be canceled out by adding r , and subtracting $\lfloor \sqrt{n} \rfloor$ twice. What we *cannot* eliminate is the “remainder” $r = n - \lfloor \sqrt{n} \rfloor^2$ because the *function* Square root cannot be inverted in \mathbb{Z} , and the algorithm cannot forget it. \blacksquare

The mkpair function. Figure 8d shows the behavior of the function `mkpair`. It is very similar to the one of `cp`, but it uses an alternative algorithm described in [4]. Here we do not describe it in detail because it’s just a composition of sums, products and square roots, which have already been discussed.

A note on the mechanization of proofs We recall once more that everything defined here above has been proved correct in `Lean`. For example, in [9], one can define as we did the `rpp` term `sqrt` and prove its behavior:

```
lemma sqrt_def (n : ℕ) (X : list ℤ) :
  <sqrt>(n::0::0::0::0::0::X) =
    n::(n-√n*√n)::(√n+√n-(n-√n*√n))::0::√n::X
```

In order to prove these theorems we make use of a *tactic* (which is a command used to build proofs) known as `simp`, which is able to automatically simplify expressions until one gets a trivial identity. What is meant by *simplify*, is that theorems which state an equality like $LHS = RHS$ (e.g. `sqrt_def`) can be marked with the attribute `@[simp]`, which means that everytime the `simp` tactic is invoked in another proof, if the equality to be proved has the expression LHS then it will be substituted with RHS , often making it simpler.

This technique is really powerful, because it makes it possible to essentially automate many proofs of theorems which in turn can be marked with `@[simp]` and be used to prove yet more theorems.

4 Proving in Lean that RPP is PRF-complete

We formally show in Lean that the class of functions we can express as (algorithms) in `rpp` contains at least PRF; so, we say “`rpp` is PRF-complete”. The definition of PRF that we take as reference is one of the two available in `mathlib` library of Lean. Once recalled and commented it briefly, we shall proceed with the main aspects of the PRF-completeness of `rpp`.

```

inductive primrec:( $\mathbb{N} \rightarrow \mathbb{N}$ )  $\rightarrow$  Prop
| zero: primrec ( $\lambda$  (n: $\mathbb{N}$ ), 0)
| succ: primrec succ
| left: primrec ( $\lambda$  (n: $\mathbb{N}$ ), (unpair n).fst)
| right: primrec ( $\lambda$  (n: $\mathbb{N}$ ), (unpair n).snd)
| pair {F G}: primrec F  $\rightarrow$  primrec G  $\rightarrow$  primrec ( $\lambda$  (n: $\mathbb{N}$ ), mkpair (F n) (G n))
| comp {F G}: primrec F  $\rightarrow$  primrec G  $\rightarrow$  primrec ( $\lambda$  (n: $\mathbb{N}$ ), F (G n))
| prec {F G}: primrec F  $\rightarrow$  primrec G  $\rightarrow$  primrec
(unpaired ( $\lambda$  (z n: $\mathbb{N}$ ), nat.rec (F z) ( $\lambda$  (y IH: $\mathbb{N}$ ), G (mkpair z (mkpair y IH))) n))
    
```

Fig. 11: `primrec` defines PRF in `mathlib` of Lean.

4.1 Primitive Recursive Functions `primrec` of `mathlib`

Figure 11 recalls the definition of PRF from [3] available in `mathlib` that we take as reference. It is an inductively defined `Proposition` `primrec` that requires a *unary* function with type $\mathbb{N} \rightarrow \mathbb{N}$ as argument. Specifically, `primrec` is the least collection of functions $\mathbb{N} \rightarrow \mathbb{N}$ with a given set of base elements, closed under some composition schemes.

Base functions. The *constant* function `zero` yields 0 on every of its inputs. The *successor* gives the natural number next to the one taken as input. The two *projections* `left`, and `right` take an argument `n`, and extract a left, or a right, component from it as `n` was the result of pairing two values $x, y: \mathbb{N}$. The functions that `primrec` relies on to encode/decode pairs on natural numbers as a single natural one are `mkpair`: $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$, and `unpair`: $\mathbb{N} \rightarrow \mathbb{N} \times \mathbb{N}$. The first one builds the value `mkpair x y`, i.e. the number of steps from the origin to reach the point with coordinates (x, y) in the path of Figure 8d. The function `unpair`: $\mathbb{N} \rightarrow \mathbb{N} \times \mathbb{N}$ takes the number of steps to perform on the same path. Once it stops, the coordinates of that point are the two natural numbers we are looking for. So, `mkpair/unpair` are an alternative to Cantor Pairing/Un-pairing.

Composition schemes. Three schemes exist in `primrec`, each depending on parameters `f, g: primrec`. The scheme `pair` builds the function that, taken a value $n: \mathbb{N}$, gives the unique value in \mathbb{N} that encodes the pair of values `F n`, and `G n`;

everything we might pack up by means of `pair`, we can unpack with `left`, and `right`.

The scheme `comp` composes `F,G:primrec`.

The *primitive recursion* scheme `prec` can be “unfolded” to understand how it works; this reading will ease the description of how to encode it in `rpp`. Let `F, G` be two elements of `primrec`. We see `prec` as encoding the function:

$$H[F,G](x) = R[G](F((x)_1), (x)_2) \quad (5)$$

where: (i) $(x)_1$ denotes `(unpair x).fst`, (ii) $(x)_2$ denotes `(unpair x).snd`, and (iii) $R[G]$ behaves as follows:

$$\begin{aligned} R[G](z, 0) &= z \\ R[G](z, n + 1) &= G(\langle z, \langle n, R[G](z, n) \rangle \rangle) , \end{aligned} \quad (6)$$

defined using the built-in recursive scheme `nat.rec` on \mathbb{N} , and $\langle a, b \rangle$ denotes `(mkpair a b)`.

4.2 The main point of the proof

In order to formally state what we mean for `rpp` to be PRF-complete, in Lean we need to say when, given `F:N → N`, we can *encode* it by means of some `f:rpp`:

```
def encode (F:N → N) (f:rpp) :=
  ∀ (z:ℤ) (n:N), <f> (z::n::repeat 0 (f.arity-2))
    = (z+(F n))::n::repeat 0 (f.arity-2)
```

says that, fixed `F:N → N`, and `f`, the statement `(encode F f)` holds if the evaluation of `<f>`, applied to any argument `(z::n::0::...::0)`, with as many occurrences of trailing 0s as `f.arity-2`, gives a list with form `((z+(F n))::n::0::...::0)` such that: (i) the first element is the original value `z` increased with the result `(F n)` of the function we want to encode; (ii) the second element is the initial `n`; (iii) the trailing 0s are again as many as `f.arity-2`. In Lean we can prove:


```
theorem completeness (F:N → N): primrec F → ∃ f:rpp, encode F f
```

which says that we know how to build `f:rpp` which encodes `F`, for every well formed `F:N → N`, i.e. such that `primrec F` holds.

The proof proceeds by induction on the proposition `primrec`, which generates 7 sub-goals. We illustrate the main arguments to conclude the most interesting case which requires to encode the composition scheme `prec`.

Remark 5. Many aspects we here detail out were simply missing in the original PRF-completeness proof for RPP in [17]. ■

The inductive hypothesis to show that we can encode `prec` is that, for any given `F,G:N → N` such that `(primrec F):Prop`, and `(primrec G):Prop`, both `f,g:rpp` exist such that `(encode F f)`, and `(encode G g)` hold. This means


 Fig. 12: Encoding `prec` of Figure 11 in `rpp`.

that $f (z :: n :: \mathbf{0}) = (z + F n) :: n :: \mathbf{0}$, and $g z :: n :: \mathbf{0} = (z + G n) :: n :: \mathbf{0}$, where $\mathbf{0}$ stands for a sufficiently long list of 0s.


Figure 12a, where we assume $z = 0$, defines `prec [f, g] : rpp` such that `(encode (prec F G) prec [f, g]) : Prop` holds, and `H [f, g]` encodes $H [F, G]$ as in (5). The term `It R [g]` in `H [f, g]` encodes (6) by iterating `R [g]` from the initial value given by `f`.

Figure 13 splits the definition of `R [g]` into three logical parts. Figure 13a packs everything up by means of `mkpair` to build the argument $R [G] (z, n)$ of `g`; by induction we get $R [G] (z, n + 1)$. In Figure 13b, `unpair` unpacks $\langle z, \langle n, R [G] (z, n) \rangle \rangle$ to expose its component to the last part. Figure 13c both increments n , and packs $R [G] (z, n)$ into s , by means of `mkpair`, because $R [G] (z, n)$ has become useless once obtained $R [G] (z, n + 1)$ from it. Packing $R [G] (z, n)$ into s , so that we can eventually recover it, is mandatory. We cannot “replace” $R [G] (z, n)$ with 0 because that would not be a reversible action.

Remark 6. The function `cp` in Figure 9e can replace `mkpair` in Figure 13c as a bijective map \mathbb{N}^2 into \mathbb{N} . Indeed, the original PRF-completeness of RPP relies on `cp`. We favor `mkpair` to take the most out of `mathlib`. ■

5 Conclusion and developments

We give a concrete example of reversible programming in a proof-assistant. We think it is a valuable operation because programming reversible algorithms is not as much wide-spread as classical iterative/recursive programming, in particular by means of a tool that allows us to certify the result. Other proof assistants have been considered, and in fact the same theorems have also been proved in `Coq`, but we found that the use of the `mathlib` library together with the `simp` tactic made our experience with `Lean` much smoother. Furthermore, our work can migrate to `Lean 4` whose stable release is announced in the near future. `Lean 4` exports

Fig. 13: Encoding $R[G]$ in (6) as $R[g] : \text{rpp}$.

its source code as efficient C code [14]; our and other reversible algorithms can become efficient extensions of Lean 4, or standalone, and C applications.

The most application-oriented obvious goal to mention is to keep developing a Reversible Computation-centered certified software stack, spanning from a programming formalism more friendly than `rpp`, down to a certified emulator of Pendulum ISA, passing through compiler, and optimizer whose properties we can certify. For example, we can also think of endowing Pendulum ISA emulators with energy-consumption models linked to the entropy that characterize the reversible algorithms we program, or the Pendulum ISA object code we can generate from them.

A more speculative direction, is to keep exploring the existence of programming schemes in `rpp` able to generate functions, other than Cantor Pairing, etc., which we can see as discrete space-filling functions, whose behavior we can describe as steps, which we count, along a path in some space.

References

1. A. Asperti, C. Sacerdoti Coen, E. Tassi, and S. Zacchiroli. User interaction with the matita proof assistant. *Journal of Automated Reasoning*, 39:109–139, Aug. 2007.
2. G. Cantor. Ein beitrag zur mannigfaltigkeitslehre. *Journal für die reine und angewandte Mathematik*, 84, 1878.

3. M. Carneiro. computability.primrec. https://leanprover-community.github.io/mathlib_docs/computability/primrec.html.
4. M. Carneiro. Formalizing computability theory via partial recursive functions. In *10th International Conference on Interactive Theorem Proving, ITP 2019, September 9-12, 2019, Portland, OR, USA*, pages 12:1–12:17, 2019.
5. L. de Moura, S. Kong, J. Avigad, F. van Doorn, and J. von Raumer. The lean theorem prover (system description). In A. P. Felty and A. Middeldorp, editors, *Automated Deduction - CADE-25*, pages 378–388, Cham, 2015. Springer International Publishing.
6. A. De Vos. *Reversible Computing - Fundamentals, Quantum Computing, and Applications*. Wiley, 2010.
7. G. Jacopini and P. Mentrasti. Generation of invertible functions. *Theor. Comput. Sci.*, 66(3):289–297, 1989.
8. G. Maletto. A Formal Verification of Reversible Primitive Permutations. BSc Thesis, Dipartimento di Matematica – Torino, October 2021. <https://github.com/GiacomoMaletto/RPP/tree/main/Tesi>.
9. G. Maletto. RPP in LEAN. <https://github.com/GiacomoMaletto/RPP/tree/main/Lean>.
10. A. Matos, L. Paolini, and L. Roversi. The fixed point problem of a simple reversible language. *TCS*, 813:143–154, 2020.
11. A. B. Matos. Linear programs in a simple reversible language. *Theor. Comput. Sci.*, 290(3):2063–2074, 2003.
12. A. B. Matos, L. Paolini, and L. Roversi. On the expressivity of total reversible programming languages. In I. Lanese and M. Rawski, editors, *Reversible Computation*, pages 128–143, Cham, 2020. Springer International Publishing.
13. K. Morita. *Theory of Reversible Computing*. Monographs in Theoretical Computer Science. An EATCS Series. Springer, 2017.
14. L. d. Moura and S. Ullrich. The Lean 4 Theorem Prover and Programming Language. In A. Platzer and G. Sutcliffe, editors, *Automated Deduction – CADE 28*, pages 625–635, Cham, 2021. Springer International Publishing.
15. L. Paolini, M. Piccolo, and L. Roversi. A certified study of a reversible programming language. In T. Uustalu, editor, *TYPES 2015 postproceedings*, volume 69 of *LIPICs*. Schloss Dagstuhl - Leibniz-Zentrum fuer Informatik, Germany, 2017.
16. L. Paolini, M. Piccolo, and L. Roversi. On a class of reversible primitive recursive functions and its turing-complete extensions. *New Generation Computing*, 36(3):233–256, July 2018.
17. L. Paolini, M. Piccolo, and L. Roversi. A class of recursive permutations which is primitive recursive complete. *Theor. Comput. Sci.*, 813:218–233, 2020.
18. K. S. Perumalla. *Introduction to Reversible Computing*. Chapman & Hall/CRC Computational Science. Taylor & Francis, 2013.
19. H. Rogers. *Theory of recursive functions and effective computability*. McGraw-Hill series in higher mathematics. McGraw-Hill, 1967.
20. M. P. Szudzik. The Rosenberg-Strong Pairing Function. *CoRR*, abs/1706.04129, 2017.